Chapter 10 Exercise Solutions





E10.1 SPI is a synchronous protocol because there is a common clock signal synchronizes the transmitter and the receiver.





E10.2 We need to program both the DDRD and the SPCR registers.  We need to configure the SS, MOSI, SCK, & TxD pins for output and MISO and RxD pins for input.  Write the value $3A into DDRD.  For the SPCR register,





-	Bit 7 (SPIE): set to 0 to disable the SPI interrupt.


-	Bit 6 (SPE): set to 1 to enable the SPI function.


-	Bit 5 (DWOM): set to 0 to select normal CMOS pin.


-	Bit 4 (MSTR): set to 1 to select master mode.


-	Bits 3 & 2 (CPOL& CPHA): set to 01 to use falling edge to shift data in and out.


-	Bits 1 & 0 (SPR1 & SPR0): set to 01 to set the data rate to be 1/4 of the frequency 


	of E clock.





The following instruction sequence will set up the desired operation parameters:





REGBAS	equ	$1000	; base address of the I/O register block


DDRD	equ	$09	; offset of DDRD from REGBAS


SPCR	equ	$28	; offset of SPCR from REGBAS


DDRD_in	equ	$3A	; value to initialize the DDRD register


SPCR_in	equ	$55	; value to initialize the SPCR register





	LDX	#REGBAS


	LDAA	#DDRD_in


	STAA	DDRD,X


	LDAA	#SPCR_in


	STAA	SPCR,X





E10.3 Connect the SPI pins as follows:


SS pin: 	configured for output and is connected to the EN pin.


SCK pin: 	configured for output and is connected to the CLK pin.


MOSI pin: 	configured for output and is connected to the Din pin.


Write the value $3A into the DDRD register to configure the desired pin directions.





When simulating the SPI transfer, the most significant bit must be sent out first.  Place the value $2B in accumulator A and use accumulator B as the loop count.





REGBAS	equ	$1000	; base address of the I/O register block


DDRD	equ	$09	; offset of DDRD from REGBAS


SPCR	equ	$28	; offset of SPCR from REGBAS


PORTD	equ	$08	; offset of PORTD from REGBAS


DDRD_in	equ	$3A	; value to initialize the DDRD register





	LDX	#REGBAS


	LDAA	#DDRD_in	; configure port D pins


	STAA	DDRD,X	;	"


	BCLR	SPCR,X $40	; disable the SPI function	


	LDAB	#6		; initialize the loop count to 6


	LDAA	#$2B		; place the data in A


	LSLA					; shift out the most significant two bits in A


	LSLA			;	"


loop6	BSET	PORTD,X $10 ; set the SCK pin to high


	LSLA		


	BCC	set0


	BSET	PORTD,X $08 ; shift out 1 from the MOSI pin


	BRA	shiftout


set0	BCLR	PORTD,X $08 ; shift out 0 from the MOSI pin


shiftout	BCLR	PORTD,X $10 ; generate a falling edge on SCK to shift the bit on MOSI


	DECB


	BNE	loop6


	...


	END





E10.4 To read a byte from the HC589 #2, we need to set the OE input of this chip to low and all other HC589s to high.  Configure the SPCR register as follows:





- Bit 7 (SPIE): set to 0 to disable SPI interrupt


- Bit 6 (SPE): set to 1 to enable the SPI function.


- Bit 5 (DWOM): set to 0 to select normal port D pin.


- Bit 4 (MSTR): set to 1 to select master mode.


- Bits 3 & 2 (CPOL& CPHA): set to 00 to use the rising edge to shift data in.


- Bits 1 & 0 (SPR1 & SPR0): set to 00 to set the data rate to be 1/2 of the frequency of E clock.





REGBAS	equ	$1000	; base address of the I/O register block


DDRD	equ	$09	; offset of DDRD from REGBAS


SPCR	equ	$28	; offset of SPCR from REGBAS


SPSR	equ	$29	; offset of SPSR from REGBAS


SPDR	equ	$2A	; offset of SPDR from REGBAS


PORTD	equ	$08	; offset of PORTD from REGBAS


PORTB	equ	$04	; offset of PORTB from REGBAS


DDRD_in	equ	$3A	; value to initialize the DDRD register


SPCR_in	equ	$50	; value to initialize the SPCR register


	


	ORG	$C000


	LDX	#REGBAS


	LDAA	#DDRD_in	; configure port D pin directions


	STAA	DDRD,X	;	"


	LDAA	#SPCR_in	; initialize the SPI system


	STAA	SPCR,X	;	"


* The following two instructions latch data into HC589s in parallel


	BCLR	PORTD,X $02 ; pull TxD (LC) pin to low


	BSET 	PORTD,X $02 ; pull TxD (LC) pin to high





	BCLR	PORTD,X $20 ; pull SS pin to  low to select parallel load mode


	BSET	PORTD,X $20 ; pull SS pin to high to select serial shift mode


	LDAA	#$FB		; enable HC589 #2 to shift data


	STAA	PORTB,X	; 	"


	STAA	SPDR,X	; trigger 8 SCK pulses sent out from SCK pin


wait	LDAA	SPSR,X	; wait until SPI transfer is completed


	BPL	wait		; 	"


	LDAA	SPDR,X	; place the byte in accumulator A


	END





E10.5 To write a byte to the HC595 #4, we need to generate a rising edge on the LC input of this


chip.  Configure the SPI as follows:





- Bit 7 (SPIE): set to 0 to disable SPI interrupt


- Bit 6 (SPE): set to 1 to enable the SPI function.


- Bit 5 (DWOM): set to 0 to select normal CMOS pin.


- Bit 4 (MSTR): set to 1 to select master mode.


- Bits 3 & 2 (CPOL& CPHA): set to 00 to use the rising edge to shift data in.


- Bits 1 & 0 (SPR1 & SPR0): set to 00 to set the data rate to be half of the E clock frequency.





REGBAS	equ	$1000	; base address of the I/O register block


DDRD	equ	$09	; offset of DDRD from REGBAS


SPCR	equ	$28	; offset of SPCR from REGBAS


SPSR	equ	$29	; offset of SPSR from REGBAS


SPDR	equ	$2A	; offset of SPDR from REGBAS


PORTD	equ	$08	; offset of PORTD from REGBAS


PORTB	equ	$04	; offset of PORTB from REGBAS


DDRD_in	equ	$3A	; value to initialize the DDRD register


SPCR_in	equ	$50	; value to initialize the SPCR register


	


	ORG	$C000


	LDX	#REGBAS


	LDAA	#DDRD_in	; configure port D pin directions


	STAA	DDRD,X	;	"


	LDAA	#SPCR_in	; initialize the SPI system


	STAA	SPCR,X	;	"


	LDAA	#$7F		; send out the byte $7F


	STAA	SPDR,X	;	"


wait	LDAA	SPSR,X	; wait until 8 bits have been shifted out


	BPL	wait		;	"


	BCLR	PORTB,X $10 ; generate a rising edge on the LC pin of HC595 #4


	BSET	PORTB,X $10 ;	"


	END





E10.6 The only thing that needs to be changed is the value to be sent to the MC14489:





REGBAS	equ	$1000	; base address of the I/O register block


DDRD	equ	$09	; offset of DDRD from REGBAS


SPCR	equ	$28	; offset of SPCR from REGBAS


SPSR	equ	$29	; offset of SPSR from REGBAS


SPDR	equ	$2A	; offset of SPDR from REGBAS


PORTD	equ	$08	; offset of PORTD from REGBAS


PORTB	equ	$04	; offset of PORTB from REGBAS


DDRD_in	equ	$3A	; value to initialize the DDRD register


SPCR_in	equ	$50	; value to initialize the SPCR register





	ORG	$00


dis_dat	FCB	$B6,$25,$32





	ORG	$C000


	LDX	#REGBAS


	LDAA	#DDRD_in


	STAA	DDRD,X


	LDAA	#SPCR_in


	STAA	SPCR,X


	BCLR	PORTD,X $20 ; pull SS pin to low to enable data shifting


	LDAA	#$01


	STAA	SPDR,X	; send out the MC14489 configuration data


	BRCLR	SPSR,X $80 *	; wait until data has been shifted out


	BSET	PORTD,X $20 ; transfer data to configuration register


	BCLR	PORTD,X $20 ; enable SPI transfer to MC14489


 	LDY	#disp_dat


	LDAB	#3


loop	LDAA	0,Y


	STAA	SPDR,X


	BRCLR	SPSR,X $80 *	; wait until data is shifted out


	INY


	DECB


	BNE	loop


	BSET	PORTD,X $20 ; transfer data to the display register


	END			 ; of the MC14489





E10.7 To display one digit at a time, we will need to use the special decode mode of the MC14489 to blank four digits and display one digit at a time. The configuration data and the display pattern is shown in Table 10S.7.





�








To create a delay of one second, the output compare function will be used.  We will perform 40 output compare operations on OC2.  Each operation creates 25 ms delay.





REGBAS	equ	$1000	; base address of the I/O register block


DDRD	equ	$09	; offset of DDRD from REGBAS


SPCR	equ	$28	; offset of SPCR from REGBAS


SPSR	equ	$29	; offset of SPSR from REGBAS


SPDR	equ	$2A	; offset of SPDR from REGBAS


PORTD	equ	$08	; offset of PORTD from REGBAS


PORTB	equ	$04	; offset of PORTB from REGBAS


TCNT	equ	$0E	; offset of TCNT from REGBAS


TOC2	equ	$18	; offset of TOC2 from REGBAS


t25ms	equ	50000	; 25 ms in number of E clock cycles


TFLG1	equ	$23	; offset of TFLG1 from REGBAS


DDRD_in	equ	$3A	; value to initialize the DDRD register


SPCR_in	equ	$50	; value to initialize the SPCR register


clr_oc2f	equ	$40	; mask to select the OC2F


OC2M	equ	$BF	; mask to select OC2F for the BCLR instruction





	ORG	$00	; table of data to be displayed


disp_tab	FCB	$DF,$81,$00,$00


	FCB	$EF,$80,$20,$00


	FCB	$F7,$80,$03,$00


	FCB	$FB,$80,$00,$40


	FCB	$FD,$80,$00,$05





	ORG	$C000


	LDX	#REGBAS


	LDAA	#DDRD_in	; set up SPI pin directions


	STAA	DDRD,X	;	"


	LDAA	#SPCR_in	; initialize the SPI function


	STAA	SPCR,X	;	"


forever	LDY	#disp_tab	; Y points to the display data


next_row	BCLR	PORTD,X $20 ; enable SPI transfer to MC14489


	LDAA	0,Y


	INY


	STAA	SPDR,X	; send data out using SPI


shift1	LDAA	SPSR,X	; wait until configuration data has been shifted out


	BPL	shift1		;	"


	BSET	PORTD,X $20	; load configuration data


	BCLR	PORTD,X $20 ; enable SPI transfer to MC14489


	LDAA	0,Y


	INY			; move the display data pointer


	STAA	SPDR,X	; send out the firs byte of display data


shift2	LDAA	SPSR,X	; wait until data has been shifted out


	BPL	shift2		; 	"


	LDAA	0,Y


	INY


	STAA	SPDR,X


	BRCLR	SPSR,X $80 *	; wait until the second display data byte is shifted out


	LDAA	0,Y


	INY


	STAA	SPDR,X


	BRCLR	SPSR,X $80 * ;	 wait until the third display data byte is shifted out


	BSET	PORTD,X $20 ; transfer data into display register





	LDAA	#clr_oc2f	; clear OC2F flag


	STAA	TFLG1,X	;	"


	LDAA	#40


	STAA	oc2_cnt	; initialize OC2 count to 40


	LDD	TCNT,X	; start OC2 operation with a delay of 25 ms


next_oc2	ADDD	#t25ms		;	"


	STD	TOC2,X	;	"


	BRCLR TFLG1,X clr_oc2f * ; wait until OC2F flag is set 


	BCLR	TFLG1,X OC2M ; clear the OC2F flag


	LDD	TOC2,X


	DEC	oc2_cnt


	BNE	next_oc2	; repeat until 1 second is expired


	CPY	#display+20	; reach the end of the display table?


	BHI	forever		; yes, start from the beginning of the display table


	BRA	next_row	; display the next row of display table


	END


			


E10.8  To display 745.23, we need to send out $01 and $B74523 as configuration and display data, respectively.


Although we are not using the SPI function, we still need to set the directions of SPI pins. We should configure SS, SCK, and MOSI pins for output.  By writing the value $3A into would achieve this objective.





REGBAS	equ	$1000	; base address of the I/O register block


DDRD	equ	$09	; offset of DDRD from REGBAS


SPCR	equ	$28	; offset of SPCR from REGBAS


PORTD	equ	$08	; offset of PORTD from REGBAS





	ORG	$C000


	LDX	#REGBAS


	LDAA	#3A


	STAA	DDRD,X ; configure SPI pin directions


	BCLR	SPCR,X $40 ; disable SPI function


	BCLR	PORTD,X $20 ; set SS pin to low to enable data shifting


	LDAA	#01	; place the configuration data in A


	LDAB	#8	; set up shift count to 8 


loop1	LSLA


	BCC	send01


	BSET	PORTD,X $08 ; output a 1 from the MOSI pin


 	BRA	clock0


send01	BCLR	PORTD,X $08 ; output a 0 from MOSI pin


clock0	BCLR	PORTD,X $10 ; 	generate a rising edge


	BSET	PORTD,X $10 ;	   “


 	DECB


	BNE	loop1


	BSET	PORTD,X $20 ; transfer configuration data


	BCLR	PORTD,X $20	; enable data shifting to MC14489


	LDAA	#$B7		; prepare to send out digits 4 and 5


	LDAB	#8		; set up loop count to 8


loop2	LSLA


	BCC	send02


	BSET	PORTD,X $08 ; output a 1 from the MOSI pin


 	BRA	clock1


send02	BCLR	PORTD,X $08 ; output a 0 from MOSI pin


clock1	BCLR	PORTD,X $10 ; 	generate a rising edge


	BSET	PORTD,X $10 ;	 "


 	DECB


	BNE	loop2


	LDAA	#$45		; prepare to send out digits 2 and 3


	LDAB	#8


loop3	LSLA


	BCC	send03


	BSET	PORTD,X $08 ; output a 1 from the MOSI pin


 	BRA	clock2


send03	BCLR	PORTD,X $08 ; output a 0 from MOSI pin


clock2	BCLR	PORTD,X $10 ; 	generate a rising edge


	BSET	PORTD,X $10 ;	 "


 	DECB


	BNE	loop3


	LDAA	#$23		; prepare to send out digits 4 and 5


	LDAB	#8		; set up loop count to 8


loop4	LSLA


	BCC	send04


	BSET	PORTD,X $08 ; output a 1 from the MOSI pin


 	BRA	clock3


send04	BCLR	PORTD,X $08 ; output a 0 from MOSI pin


clock3	BCLR	PORTD,X $10 ; 	generate a rising edge


	BSET	PORTD,X $10 ;	 "


 	DECB


	BNE	loop4


	BSET	PORTD,X $20 ; load data into the display register


	END





E10.9 The circuit connection between the 68HC11 and the 68HC68T1 is identical to Figure 10.37 and hence won't be repeated here.  The circuit connection between the 68HC11 and the MC145000 and two MC145001s is shown in Figure S9.1. The time will be displayed using


the following format :





	hh mm ss ww MM dd yy


	


where hh mm ss ww MM dd yy stand for hour, minute, second, day of week, month, date, and year respectively. 





�





For the display part,





The system clock is generated internally by MC145000 and becomes an input to two MC145001s.


The master LCD driver is responsible for driving hh mm ss while two slaves are responsible for driving MM dd ww yy and hence yy should be sent out first and hh should be send out last.


Two LCD digits are not used and hence are not connected to the rightmost MC145001.


The data clock inputs of all LCD drivers are tied to the SCK output of the 68HC11.


The LXD 8-digit LCDs are used in the circuit.





The operation of the circuit:





Part 1.  The setup of the SPI system is:





Configure SS, SCK, and MOSI pins for output (write the byte %00111010 into DDRD).


Disable SPI interrupt


Enable the SPI


Normal port D pins


Select the falling edge of SCK to shift data into the LCD drivers and the 68HC68T1.


Select 1 Mbits/sec data rate at 2 MHz E clock frequency.


Write the value $54 into the SPCR register to set the parameters described in steps 2 to 6.





Part 2. Initialization of time of day and the alarm time.  





The time of day and the alarm time should be entered into the memory using the interrupt-driven input method.  The input device can be an 8-DIP switch.  The 24-hour format is used in this example.  We will assume these values have been stored in memory locations $00-$09.  





Part 3. Wait loop.  The 68HC11 executes a main program that perform the initialization described in part 1 and 2 and initializes a 5-minute counter to 300 and then stay in an infinite loop to wait for interrupts from the 68HC68T1.  The 5-minute counter is mainly used to turn on the alarm for exactly five minutes.  The main program will also display the current time before staying in the infinite loop.





Part 4.  Interrupt service routine.  This routine performs the following operations:





Update the time display.


Check the interrupt source by reading the status register.


Simply return if the interrupt is caused by the periodic interrupt and alarm is not turned on.


Turn on the alarm if the alarm is caused by the alarm function.


Decrement the alarm count by 1 and returns if the alarm has been turned on.





The program is as follows:





regbas	equ	$1000		; base address of I/O register block


spdr	equ	$2A		; offset of spdr from regbas


spcr	equ	$28		; offset of spcr from regbas


spsr	equ	$29		; offset of spsr from regbas


ddrd	equ	$09		; offset of ddrd from regbas


portd	equ	$08		; offset of portd from regbas


ram_addr	equ	$80		; starting address to write the RAM





second_r	equ	$20		; read address for seconds


minute_r	equ	$21		; read address for minutes


hours	equ	$22		; read address for hours


day_wk_r	equ	$23		; read address for day of week


date_m_r	equ	$24		; read address for date of month


month_r	equ	$25		; read address for months


year_r	equ	$26		; read address for years


stat_reg	equ	$30		; address for status register


clk_ctrl	equ	$31		; read address for clock control register


int_ctrl	equ	$32		; read address for interrupt control register


second_w	equ	$A0		; write address for seconds


minute_w	equ	$A1		; write address for minutes


hour_w	equ	$A2		; write address for hours


day_wk_w	equ	$A3		; write address for day of week


date_m_w	equ	$A4		; write address for date of month


month_w	equ	$A5		; write address for months


year_w	equ	$A6		; write address for years


s_alarm	equ	$A8		; write address for seconds alarm


m_alarm	equ	$A9		; write address for minutes alarm


h_alarm	equ	$AA		; write address for hours alarm


clk_ctrlw	equ	$AB		; write address for clock control register


int_ctrlw	equ	$AC		; write address for interrupt control register


five_min	equ	300		; alarm count for five minutes





clk_mgt	org	$00		; block of memory locations that stores time and alarm info


ss	rmb	1		; entered by the user


min	rmb	1		;	"


hh	rmb	1		;	"


ww	rmb	1		;	"


dd	rmb	1		;	"


mm	rmb	1		;	"


yy	rmb	1		;	"





display	org	$10		; memory locations to hold time display data


y1	rmb	1		; lower digit of years


y2	rmb	1		; upper digit of years


d1	rmb	1		; lower digit of date of months


d2	rmb	1		; upper digit of date of months


mo1	rmb	1		; lower digit of months


mo2	rmb	1		; upper digit of months


wd1	rmb	1		; lower digit of day of weeks


wd2	rmb	1		; upper digit of day of weeks


s1	rmb	1		; lower digit of seconds


s2	rmb	1		; upper digit of seconds


mi1	rmb	1		; lower digtit of minutes


mi2	rmb	1		; upper digit of minutes


h1	rmb	1		; lower digit of hours


h2	rmb	1		; upper digit of hourss


alarm_s	rmb	1	;	"


alarm_m 	rmb	1			;	"


alarm_h	rmb	1		;	"


alarmcnt	rmb	2		; five-minute alarm count 


alarmflg	rmb	1		; alarm flag indicating if the alarm is on





	org	$FFF2		; set up IRQ vector


	FDB	clock_hnd	;	"





****************************************************************************


* use the following directives to replace the previous two directives if this program is to be run


* on a EVB board


*	org	$00EE


*	JMP	clock_hnd


****************************************************************************


	org	$c000		; starting address of the program


	sei			; disable interrupt to the 68HC11


	lds	#$DFFF	; initialize the stack pointer


	ldx	#regbas


	ldaa	#$3A		; configure SPI pins directions


	staa	ddrd,X		;	"


	ldaa	#$54		; initialize the SPI system


	staa	spcr,X		;	"


* The following instructions send the time-of-day information to the 68HC68T1 in one burst


	bset	portd,X $20	; enable SPI transfer to the 68HC68T1


	ldaa	#second_w	; send the seconds write address to the 68HC68T1


	staa	spdr,X		;	"


	brclr	spsr,X $80 *	; wait until the SPI transfer is completed


	ldab	#7		; set up byte count to 7


	ldy	#clk_mgt	; load the base address of time of day information


loop1	ldaa	0,Y		; send time of day to the 68HC68T1


	staa	spdr,X		;	"


	brclr	spsr,X $80 *	;	"


	iny			; move the time pointer


	decb			; decrement the byte count


	bne	loop1		


	bclr	portd,X $20	; disable SPI transfer to the 68HC68T1





* The following instruction sequence sends the alarm time to the 68HC68T1 in one burst





	bset	portd,X $20	; enable SPI transfer to the 68HC68T1


	ldaa	#s_alarm	; send the seconds alarm address to the 68HC68T1


	staa	spdr,X		;	"


	brclr	spsr,X $80 *	; 	"


	ldab	#3		; initialize the transfer byte count to 3


	ldy	#alarm_s	; load the base address of the alarm information


loop2	ldaa	0,y		; send alarm information to the 68HC68T1


	staa	spdr,X		;	"


	brclr	spsr,X $80 *	;	"


	iny			; move the pointer


	decb			; decrement the transfer byte count


	bne	loop2


	bclr	portd,X #$20	; disable SPI transfer to the 68HC68T1





* The following instruction sequence send out 16 zeroes to blank the LCDs 


	ldab	#16		; send out zeroes to blank LCDs


	clra			;	"


loop3	staa	spdr,X		;	"


	brclr	spsr,X $80 *	;	"


	decb			;	"


	bne	loop3		;	"


* The following instruction sequence initialize the timer clock control and interrupt control


* registers (done in a burst because they are located in sequential addresses)


	bset	portd,X $20	; enable SPI transfer to the 68HC68T1


	ldaa	#clk_ctrlw	; send clock control register write address to the 68HC68T1


	staa	spdr,X		;	"


	brclr	spsr,X $80 *	;	"


	ldaa	#%10110100	; initialize the timer clock control register


	staa	spdr,X		;	"


	brclr	spsr,X $80 *	;	"


	ldaa	#%00011100	; initialize the timer interrupt control register


	staa	spdr,X		;	"


	brclr	spsr,X $80 *	;	"


	bclr	portd,X $20	; disable SPI transfer to the 68HC68T1


	jsr	show_time	; call the subroutine to display the current time


	ldd	#five_min	; initialize the alarm count to 300


	std	alarmcnt	;	"


	clr	alarmflg	; clear the alarm flag


	cli			; enable interrupt to the 68HC11





forever	bra	forever		; infinite loop





* The following subroutine reads the current time from the 68HC68T1 and sends it to LCDs for


* display





show_time	bset	portd,X $20	; enable SPI transfer to the 68HC68T1


	ldaa	#second_r	; send the seconds read address to the 68HC68T1


	staa	spdr,X		; 	"			


	brclr	spsr,X $80 *	;	"


	staa	spdr,X		; read the seconds


	brclr	spsr,X $80 *	;	"


	ldab	spdr,X		;	"


	stab	temp		; save a copy in temp


	andb	#$0F		; mask out the upper seconds digit


	ldy	#lcd_dp	; compute the address of the LCD pattern of the lower 


	aby			; seconds digit


	ldaa	0,Y		; save the pattern of the lower seconds digit


	staa	s1		;	"


	ldab	temp		; get the second digits


	lsrb			; shift the upper seconds digit to B's lower four bits


	lsrb			;	"


	lsrb			;	"


	lsrb			;	"


	ldy	#lcd_ndp	; compute the address of the LCD pattern of the upper


	aby			; seconds digit


	ldab	0,Y		; save the pattern in upper seconds digit


	stab	s2		;	"


	staa	spdr,X		; read the minute digits


	brclr	spsr,X $80 *	;	"


	ldab	spdr,X		;	"


	stab	temp		; save a copy in temp


	andb	#$0F		; mask out the upper minutes digit


	ldy	#lcd_dp	; compute the address of the LCD pattern of the lower 


	aby			; minutes digit


	ldab	0,Y		; save the pattern of the lower minutes digit


	stab	mi1		;	"


	ldab	temp		; get the minute digits


	lsrb			; shift the upper minutes digit to B's lower four bits


	lsrb			;	"


	lsrb			;	"


	lsrb			;	"


	ldy	#lcd_ndp	; compute the address of the LCD pattern of the upper


	aby			; minutes digit


	ldab	0,Y		; save the pattern in upper minutes digit


	stab	mi2


	staa	spdr,X		; read the hours digits


	brclr	spsr,X $80 *	;	"


	ldab	spdr,X		;	"


	stab	temp		; save a copy in temp


	andb	#$0F		; mask out the upper hours digit


	ldy	#lcd_dp	; compute the address of the LCD pattern of the lower 


	aby			; hours digit


	ldaa	0,Y		; save the pattern of the lower hour sdigit


	staa	h1		;	"


	ldab	temp		; get the hours digits


	lsrb			; shift the upper hours digit to B's lower four bits


	lsrb			;	"


	lsrb			;	"


	lsrb			;	"


	ldy	#lcd_ndp	; compute the address of the LCD pattern of the upper


	aby			; hours digit


	ldab	0,Y		; save the pattern in upper hours digit


	stab	h2


	staa	spdr,X		; read the day-of-weeks  digits


	brclr	spsr,X $80 *	;	"


	ldab	spdr,X		;	"


	stab	temp		; save a copy in temp


	andb	#$0F		; mask out the upper day-of-weeks digit


	ldy	#lcd_dp	; compute the address of the LCD pattern of the lower 


	aby			; day-of-weeks digit


	ldaa	0,Y		; save the pattern of the lower day-of-week digit


	staa	wd1		;	"


	ldab	temp		; get the day-of-weeks digits


	lsrb			; shift the upper day-of-weeks digit to B's lower four bits


	lsrb			;	"


	lsrb			;	"


	lsrb			;	"


	ldy	#lcd_ndp	; compute the address of the LCD pattern of the upper


	aby			; day-of-week digit


	ldab	0,Y		; save the pattern in upper day-of-week digit


	stab	wd2		;	"


	staa	spdr,X		; read the date-of-month digits


	brclr	spsr,X $80 *	;	"


	ldab	spdr,X		;	"


	stab	temp		; save a copy in temp


	andb	#$0F		; mask out the upper date-of-month digit


	ldy	#lcd_dp	; compute the address of the LCD pattern of the lower 


	aby			; date-of-month digit


	ldaa	0,Y		; save the pattern of the lower date-of-month digit


	staa	d1		;	"


	ldab	temp		; get the date-of-month digits


	lsrb			; shift the upper date-of-months digit to B's lower four bits


	lsrb			;	"


	lsrb			;	"


	lsrb			;	"


	ldy	#lcd_ndp	; compute the address of the LCD pattern of the upper


	aby			; date-of-month digit


	ldab	0,Y		; save the pattern in upper date-of-month digit


	stab	d2		;	"


	staa	spdr,X		; read the month digits


	brclr	spsr,X $80 *	;	"


	ldab	spdr,X		;	"


	stab	temp		; save a copy in temp


	andb	#$0F		; mask out the upper months digit


	ldy	#lcd_dp	; compute the address of the LCD pattern of the lower 


	aby			; months digit


	ldaa	0,Y		; save the pattern of the lower months digit


	staa	mo1		;	"


	ldab	temp		; get the month digits


	lsrb			; shift the upper months digit to B's lower four bits


	lsrb			;	"


	lsrb			;	"


	lsrb			;	"


	ldy	#lcd_ndp	; compute the address of the LCD pattern of the upper


	aby			; months digit


	ldab	0,Y		; save the pattern in upper months digit


	stab	mo2		;	"


	staa	spdr,X		; read the year digits


	brclr	spsr,X $80 *	;	"


	ldab	spdr,X		;	"


	stab	temp		; save a copy in temp


	andb	#$0F		; mask out the upper years digit


	ldy	#lcd_dp	; compute the address of the LCD pattern of the lower 


	aby			; years digit


	ldaa	0,Y		; save the pattern of the lower years digit


	staa	y1		;	"


	ldab	temp		; get the year digits


	lsrb			; shift the upper years digit to B's lower four bits


	lsrb			;	"


	lsrb			;	"


	lsrb			;	"


	ldy	#lcd_ndp	; compute the address of the LCD pattern of the upper


	aby			; years digit


	ldab	0,Y		; save the pattern in upper years digit


	stab	y2		;	"


	bclr	portd,X $20	; disable SPI transfer to 68HC68T1


	ldab	#14		; set up transfer count


	ldy	#display


next	ldaa	0,Y		; send out a digit to LCD


	staa	spdr,X		;	"


	brclr	spsr,X $80 *	;	"


	iny			; move the pointer to the display data


	decb


	bne	next		; not done yet?


	rts			





clock_hnd	bset	portd,X $20	; enable SPI transfer to the 68HC68T1


	ldaa	#stat_reg	; send the status register address to the 68HC68T1


	staa	spdr,X		;	"


	brclr	spsr,X $80 *	;	"


	staa	spdr,X		; read in the status register of the 68HC68T1


	brclr	spsr,X $80 *	;	"


	bclr	portd,X $20	; disable SPI transfer to the 68HC68T1


	ldaa	spdr,X		;	"


	anda	#$02		; check the alarm interrupt flag


	beq	chkalarm	; interrupt is periodic, go and check the alarm flag


	bset	portd,X $20	; enable SPI transfer to the 68HC68T1


	ldaa	#clk_ctrlw	; send clock control register write address to the 68HC68T1


	staa	spdr,X 		;	"


	brclr	spsr,X $80 *	;	"


	ldaa	#%10110011	; send out a new control byte to enable the CLK OUT 


	staa	spdr,X		; signal to turn on the alarm and set the frequency to 4 KHz


	brclr	spsr,X $80 *	;	"


	bclr	portd,X $20	; disable SPI transfer to the 68HC68T1


	ldaa	#1		; set the alarm flag


	staa	alarmflg	;	"


	bra	exit		; prepare to return from interrupt


chkalarm	ldaa	alarmflg	; check the alarm flag


	beq	exit		; alarm is not turned on and return 


	ldd	alarmcnt	; decrement the alarm count by 1


	subd	#1		;	"


	bne	exit1		; if alarm count is not zero then return


	ldd	#five_min	; re-initialize the alarm count


	std	alarmcnt	;	"


	clr	alarmflg	; clear alarm flag to 0


	bra	exit


exit1	std	alarmcnt


exit	jsr	show_time


	rti		





lcd_dp	fcb	$FB,$70,$D7,$F5,$7C,$BD,$3F,$F0,$FF,$FC,$FE,$1A,$7A,$7B,$14


	fcb	$10


lcd_ndp	fcb	$EB,$60,$C7,$E5,$6C,$AD,$2F,$E0,$EF,$EC,$EE,$0A,$6A,$6B,$04


	fcb	$00


	END





E10.10 The circuit is shown Figure 10S.10.  

The procedure is as follows:





Step 1: Send a 4-byte configuration data ($01000001) to these two MC14489s.


Step 2. Send out the display data in this order: $856789801234.





The program is as follows:


regbas	equ	$1000		; base address of I/O register block


spdr	equ	$2A		; offset of spdr from regbas


spcr	equ	$28		; offset of spcr from regbas


spsr	equ	$29		; offset of spsr from regbas


ddrd	equ	$09		; offset of ddrd from regbas


portd	equ	$08		; offset of portd from regbas


sp_dir	equ	$3A		; value to set SPI pins diirections


spcr_ini	equ	$56		; value to initialize the SPI control register





	org	$00


conf_dat	fcb	$01,$00,$00,$01


disp_dat	fcb	$85,$67,$89,$80,$12,$34 	; data to be displayed





	org	$C000


	ldx	#regbas


* send out configuration to two MC14489s 


	bclr	portd,X $20	; enable the SPI transfer to the MC14489


	ldab	#4		; need to send three bytes to the left MC14499


loop1	ldy	#conf_dat


	ldaa	0,Y		; get one byte to send out


	staa	spdr,X	;	  “


	iny


	brclr	spsr,X $80 *	; 	"


	decb


	bne	loop1		; not done yet?


	bset 	portd,X $20	; load data into configuration register





* The following instructions send out $856789801234 to two MC14489s





	bclr	portd,X $20	; enable the SPI transfer to the MC14499s


	ldab	#6		; set up loop count to 5


	ldy	#disp_dat	; set Y to point to the display data


loop2	ldaa	0,Y		; send out one byte to the MC14499


	staa	spdr,X		;	"


	brclr	spsr,X $80 *	;	"


	iny			; move the data pointer


	decb			; decrement the loop count


	bne	loop2		; not done yet?


	bset	portd,X $20	; load data into the display register


	END


�PAGE  �10�


9-











